#-swank
(unless (member :child-sbcl *features*)
  (quit
   :unix-status
   (process-exit-code
    (run-program *runtime-pathname*
                 `("--control-stack-size" "128MB"
                   "--noinform" "--disable-ldb" "--lose-on-corruption" "--end-runtime-options"
                   "--eval" "(push :child-sbcl *features*)"
                   "--script" ,(namestring *load-pathname*))
                 :output t :error t :input t))))
;; -*- coding: utf-8 -*-
(eval-when (:compile-toplevel :load-toplevel :execute)
  (sb-int:defconstant-eqx OPT
    #+swank '(optimize (speed 3) (safety 2))
    #-swank '(optimize (speed 3) (safety 0) (debug 0))
    #'equal)
  #+swank (ql:quickload '(:cl-debug-print :fiveam) :silent t)
  #-swank (set-dispatch-macro-character
           ;; enclose the form with VALUES to avoid being captured by LOOP macro
           #\# #\> (lambda (s c p) (declare (ignore c p)) `(values ,(read s nil nil t)))))
#+swank (cl-syntax:use-syntax cl-debug-print:debug-print-syntax)
#-swank (disable-debugger) ; for CS Academy

;; BEGIN_INSERTED_CONTENTS
(defmacro dbg (&rest forms)
  #+swank
  (if (= (length forms) 1)
      `(format *error-output* "~A => ~A~%" ',(car forms) ,(car forms))
      `(format *error-output* "~A => ~A~%" ',forms `(,,@forms)))
  #-swank (declare (ignore forms)))

(defmacro define-int-types (&rest bits)
  `(progn
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "UINT~A" b)) () '(unsigned-byte ,b))) bits)
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "INT~A" b)) () '(signed-byte ,b))) bits)))
(define-int-types 2 4 7 8 15 16 31 32 62 63 64)

(declaim (inline println))
(defun println (obj &optional (stream *standard-output*))
  (let ((*read-default-float-format* 'double-float))
    (prog1 (princ obj stream) (terpri stream))))

(defconstant +mod+ 1000000007)

;;;
;;; Body
;;;

(defun calc-path (k &rest nodes)
  (sb-int:named-let recur ((x 0) (y 0) (nodes nodes) (rem-step 0))
    (cond ((zerop rem-step)
           (cons (cons x y)
                 (if nodes
                     (recur x y nodes k)
                     nil)))
          ((null nodes) nil)
          (t
           (destructuring-bind (node-x . node-y) (car nodes)
             (assert (or (= node-x x) (= node-y y)))
             (let* ((dir-x (= node-y y))
                    (signed-dist (if dir-x
                                     (- node-x x)
                                     (- node-y y)))
                    (sign (signum signed-dist)))
               (if (> (abs signed-dist) rem-step)
                   (if dir-x
                       (recur (+ x (* sign rem-step)) y nodes 0)
                       (recur x (+ y (* sign rem-step)) nodes 0))
                   (recur node-x node-y (cdr nodes) (- rem-step (abs signed-dist))))))))))

(defun solve (x y k)
  (if (zerop (mod (+ x y) k))
      (calc-path k (cons 0 y) (cons x y))
      (let* ((ideal-n (ceiling (+ x y) k))
             (n (if (= ideal-n 1)
                    (if (oddp (- (* 2 k) (+ x y)))
                        (if (oddp k)
                            3
                            (return-from solve))
                        2)
                    (if (oddp (- (* ideal-n k) (+ x y)))
                        (if (oddp k)
                            (+ ideal-n 1)
                            (return-from solve))
                        ideal-n))))
        (if (and (= ideal-n 1) (= n 3))
            (let* ((extra (/ (- (* 3 k) (+ x y)) 2))
                   (extra1 (floor extra 2))
                   (extra2 (ceiling extra 2)))
              (calc-path k (cons (- extra1) 0)
                         (cons (- extra1) (+ y extra2))
                         (cons x (+ y extra2))
                         (cons x y)))
            (let ((extra (/ (- (* n k) (+ x y)) 2)))
              (check-type extra fixnum)
              (calc-path k (cons 0 (+ y extra))
                         (cons x (+ y extra))
                         (cons x y)))))))

(defun main ()
  (let* ((k (read))
         (x (read))
         (y (read))
         (sign-x (>= x 0))
         (sign-y (>= y 0))
         (tx (abs x))
         (ty (abs y))
         (swap-p (< tx ty)))
    (when swap-p
      (rotatef tx ty))
    (let ((res (cdr (solve tx ty k))))
      (unless res
        (println -1)
        (return-from main))
      (println (length res))
      (dolist (node res)
        (let ((x (car node))
              (y (cdr node)))
          (when swap-p (rotatef x y))
          (unless sign-x (setq x (- x)))
          (unless sign-y (setq y (- y)))
          (format t "~D ~D~%" x y))))))

#-swank (main)

;;;
;;; Test and benchmark
;;;

#+swank
(defun io-equal (in-string out-string &key (function #'main) (test #'equal))
  "Passes IN-STRING to *STANDARD-INPUT*, executes FUNCTION, and returns true if
the string output to *STANDARD-OUTPUT* is equal to OUT-STRING."
  (labels ((ensure-last-lf (s)
             (if (eql (uiop:last-char s) #\Linefeed)
                 s
                 (uiop:strcat s uiop:+lf+))))
    (funcall test
             (ensure-last-lf out-string)
             (with-output-to-string (out)
               (let ((*standard-output* out))
                 (with-input-from-string (*standard-input* (ensure-last-lf in-string))
                   (funcall function)))))))

#+swank
(defun get-clipbrd ()
  (with-output-to-string (out)
    (run-program "C:/msys64/usr/bin/cat.exe" '("/dev/clipboard") :output out)))

#+swank (defparameter *this-pathname* (uiop:current-lisp-file-pathname))
#+swank (defparameter *dat-pathname* (uiop:merge-pathnames* "test.dat" *this-pathname*))

#+swank
(defun run (&optional thing (out *standard-output*))
  "THING := null | string | symbol | pathname

null: run #'MAIN using the text on clipboard as input.
string: run #'MAIN using the string as input.
symbol: alias of FIVEAM:RUN!.
pathname: run #'MAIN using the text file as input."
  (let ((*standard-output* out))
    (etypecase thing
      (null
       (with-input-from-string (*standard-input* (delete #\Return (get-clipbrd)))
         (main)))
      (string
       (with-input-from-string (*standard-input* (delete #\Return thing))
         (main)))
      (symbol (5am:run! thing))
      (pathname
       (with-open-file (*standard-input* thing)
         (main))))))

#+swank
(defun gen-dat ()
  (uiop:with-output-file (out *dat-pathname* :if-exists :supersede)
    (format out "")))

#+swank
(defun bench (&optional (out (make-broadcast-stream)))
  (time (run *dat-pathname* out)))

;; To run: (5am:run! :sample)
#+swank
(it.bese.fiveam:test :sample
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "11
-1 2
"
    "3
7 4
2 10
-1 2
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "4600
52 149
"
    "-1
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "4
9 9
"
    "5
1 3
4 2
4 6
6 8
9 9
")))


Measured memory_bytes: 41912